Numerical Implementation of the HHP Decomposition in GAMS Models - An Illustrative Example

Ref.: **Who should pay how much? - Compensation for International Spillovers from Carbon Abatement Policies to Developing Countries - A Global CGE Assessment**

Authors: C. Böhringer (boehringer@zew.de) and T.F. Rutherford (rutherford@colorado.edu)

We illustrate the numerical implementation of the HHP decomposition along a simple multi-region exchange model. In each region a representative consumer is endowed with a macro-good which either can be either domestically used or exported. Domestic consumption is captured by a constant-elasticity-of-substitution (CES) cost function which characterizes the substitution possibilities between the domestic macro good and an import composite (*zprf_c*)[1]:

$$p_r^C \text{ - } \left(\theta_d \, {p_r}^{1-\sigma_{dm}} + (1-\theta_d)\, {p_r^M}^{1-\sigma_{dm}}\right)^{\frac{1}{1-\sigma_{dm}}} = 0 \qquad \perp C_r$$

where:

p_r^C is the consumption price index for region *r*,

θ_d is the base year value share of the domestic good in total consumption of region *r*,

p_r is the macro good price in each region *r*,

p_r^M is the price index for the import composite price index of region *r*,

σ_{dm} is the substitution elasticity between the domestic macro good and the import aggregate,

and

C_r is the associated activity level of consumption.

The import composite is a CES aggregate of macro good imports from other regions on which an import tariff may be imposed (*zprf_m*):

$$p_r^M \text{ - } \left(\sum_{rr \neq r} \theta_{rr,r} \left(p_{rr}(1+tm_r)\right)^{1-\sigma_{mm}}\right)^{\frac{1}{1-\sigma_{mm}}} = 0 \qquad \perp M_r$$

[1] We add the names of the corresponding equations in the model source code within parenthesis.

where:

$\theta_{rr,r}$ is the base year value share of the macro good from region rr in the import aggregate of region *r*,

tm_r is the tariff rate of region *r*,

σ_{mm} is the substitution elasticity between imports from different regions,

and

M_r is the associated activity level of import composite formation.

Market clearance for the domestic macro good in each region implies (*mkt_p*):

$$gdp_r = c0_{r,r} C_r \left(\frac{p_r^C}{p_r} \right)^{\sigma_{dm}} + \sum_{rr \neq r} c0_{r,rr} M_{rr} \left(\frac{p_{rr}^M}{p_r (1 + tm_{rr})} \right)^{\sigma_{mm}} \qquad \perp p_r$$

where:

gdp_r is the endowment with the macro good for region *r*,

$c0_{r,rr}$ is the base year trade flow from region *r* to region *rr*,

and

p_r is the associated macro good price in each region *r*.

The supply-demand balance for the import composite in each region can be written as (*mkt_pm*):

$$m0_r M_r = m0_r C_r \left(\frac{p_r^C}{p_r^M} \right)^{\sigma_{dm}} \qquad \perp p_r^M$$

where:

$m0_r$ is the aggregate base year import to region *r* from all other regions,

and

p_r^M is the associated price index for the import composite of region *r*.

Finally, aggregate consumption in each region is determined by the available income of the representative agent which includes potential tariff revenues (*mkt_pc*):

$$gdp_r C_r p_r^C = gdp_r p_r + \sum_{rr \neq r} c0_{rr,r} M_r \left(\frac{p_r^M}{p_{rr} (1 + tm_r)} \right)^{\sigma_{mm}} tm_r p_{rr} \qquad \perp p_r^C$$

where:

p_r^C is the associated consumption price index of region *r*,

The competitive equilibrium for this small exchange equilibrium is characterized by the above zero-profit conditions and market clearance conditions. In equilibrium, each zero-profit condition is linked to a activity level, and each market clearance condition is linked to a price level.

Below, we provide the source code for our maquette written in GAMS. We start from a base year (benchmark) equilibrium where no tariffs apply. In the policy counterfactual each region levies a uniform import tariff of 10 %. The decomposition subroutine delivers the information how the total welfare change for each region (measured in terms of real consumption) can be attributed to the changes in policy instruments (here: tariffs) across regions.

```
$TITLE Simple Multi-Region Exchange Model Illustrating Decomposition Method

set r   Regions /r1*r3/;

alias (r,rr,s);

parameters  gdp(r)      GDP Index   /r1 1, r2 2, r3 3/,
            tm(r)       Tariff rates,
            m0(r)       Base year aggregate trade (excl. intra-region trade);

*   Base year tariff is zero
tm(r)       = 0;

table   c0(r,rr)    Base year trade flows (incl. intra-region trade)
            R1          R2          R3
R1       0.167       0.333       0.500
R2       0.333       0.667       1.000
R3       0.500       1.000       1.500;

m0(r) = sum(rr$(not sameas(r,rr)), c0(rr,r));

parameters
    theta_d(r)      Value share of domestic good in total consumption of region r,
    theta_m(rr,r)   Value share of good from region rr in import composite of region r,
    esub_dm(r)     Substitution elasticity between domestic good and import composite,
    esub_mm(r)      Substitution elasticity between different imports;

theta_d(r) = c0(r,r)/gdp(r);
theta_m(rr,r)$(not sameas(r,rr)) = c0(rr,r)/m0(r);
esub_dm(r) = 2;
esub_mm(r) = 4;


variables
    m(r)    Import composite formation activity level,
    c(r)    Consumption demand,
    p(r)    Macro output,
    pc(r)   Consumption price,
```

```
    pm(r)   Import composite price index;

equations
    zprf_c(r)       Zero profit condition for consumption formation,
    zprf_m(r)       Zero profit condition for import composite formation,
    mkt_p(r)        Market clearance for exchange good,
    mkt_pm(r)       Market clearance for exchange good,
    mkt_pc(r)       Market clearance for aggregate consumption good;

*   Zero profit per unit activity
*   =============================

*   Production:
zprf_c(r).. {theta_d(r)*p(r)**(1-esub_dm(r)) + (1-theta_d(r))*pm(r)**(1-esub_dm(r))
                }**(1/(1-esub_dm(r)))
                =e= pc(r);

*   Import-domestic aggregation:
zprf_m(r)..     sum(rr$(not sameas(r,rr)),theta_m(rr,r)*(p(rr)*(1+tm(r)))**
                    (1-esub_mm(r)))**(1/(1-esub_mm(r)))  =e= pm(r);

*   Market clearing
*   ===============

*   Import aggregate:
mkt_pm(r)..     m(r)*m0(r) =e=  m0(r)*c(r)*(pc(r)/pm(r))**esub_dm(r);

*   Domestic macro good:
mkt_p(r)..      gdp(r)  =e=  c0(r,r)*c(r)*(pc(r)/p(r))**esub_dm(r)
                                + sum(rr$(not sameas(r,rr)), m(rr)*c0(r,rr)*
                                (pm(rr)/(p(r)*(1+tm(rr))))**esub_mm(rr) );

*   Domestic consumption good:
mkt_pc(r)..     c(r)*gdp(r)*pc(r) =e= p(r)*gdp(r) + sum(rr$(not sameas(r,rr)),
                                       tm(r)*p(rr)*m(r)*c0(rr,r)*
                                             (pm(r)/(p(rr)*(1+tm(r))))**esub_mm(r));


*   Define the mcp model:

model exchange /zprf_c.c, zprf_m.m, mkt_p.p, mkt_pm.pm, mkt_pc.pc/;



*   Install lower bounds to avoid bad function calls:
pc.lo(r)    = 1e-5;
pm.lo(r)    = 1e-5;
p.lo(r)     = 1e-5;

*   Install level values:
c.l(r)=1; m.l(r)=1; p.l(r)=1; pc.l(r)=1; pm.l(r)=1;

*   Replicate the benchmark
exchange.iterlim = 0;
solve exchange using mcp;

display "Benchmark tolerance:", exchange.objval;

*   Reset iteration limit for policy counterfactuals
exchange.iterlim = 2000;


* ------------------------------------------------------------------
$stitle  GAMS Code for Decomposition
```

```
set     t               Steps in line integral /t0*t10/;

parameter
        tm0(r)          Base value of policy instrument
        deltatm(r)      Change in policy instrument
        w0(r,t)         Value of W along integral over t
        dwdtm(r,t,r)    Value of partial derivatives along integral
        epsilon         Differentiation perturbation
        decomp(r,r)     Decomposition of changes in w
        pctdecomp(r,r)  Decomposition in percentage effects
        handshake       Approximation error
        dt              Step size in integral over t;

*  Step size for the line integral:

dt = 1 /(card(t)-1);

*   Define a reasonable value for numerical differencing:

epsilon = 0.0001;
tm0(r) = 0;
deltatm(r) = 0.1;

tm(r) = tm0(r);
loop(t,
        solve exchange using mcp;
        w0(r,t) = c.l(r);
        loop(s,
                tm(s) = tm(s) + epsilon;
        solve exchange using mcp;
                dwdtm(r,t,s) = (c.l(r) - w0(r,t)) / epsilon;
                tm(s) = tm(s) - epsilon;
        );
        tm(r) = tm(r) + dt * deltatm(r);
);

decomp(r,rr) = sum(t, dwdtm(r,t,rr) * deltatm(rr) * dt);
handshake(r) = sum(rr, decomp(r,rr)) - sum(t$(ord(t) gt 1), w0(r,t) - w0(r,t-1));


pctdecomp(r,rr) = round( 100 * decomp(r,rr) / sum(s, decomp(r,s)));
option pctdecomp:0;

display decomp, handshake, pctdecomp;
```